Differential Independence of Solutions of a Class of q-Hypergeometric Difference Equations

Charlotte Hardouin, Michael F. Singer
REVISED November 6, 2012

This note is a companion to the paper "Differential Galois Theory of Difference Equations" by the same authors. In a previous version of this note (and in Example 3.14 of this paper) we claimed that for the q-hypergeometric equation with prarameters restricted as below, the associated σδ-PV extension has differential transcendence degree 3. We have since founds gaps in the proof of this claim. Nonetheless, we are able to show this claim when we further restrict the papramaters. In the following we show how MAPLE can be used to verify these cases.

We shall consider the equations

$$y(q^2x) -[(a+b)x-(1+c/q))/(abx -c/q)]y(qx) + [(x-1)/ (abx -c/q)]y(x) = 0$$

where $a = b$, $c=q$, $a\notin q^{\mathbb{Z}}$, $a^2 \in q^{\mathbb{Z}}$. We shall show in the cases when $a^2 = q, q^3, q^5, q^{-1}, q^{-3}, q^{-5}$ and when $a^2 = q^2$, $a \neq q$ (that is, $a=-q$), the associated $-$PV extension has differential transcendence degree 3. The MAPLE code we give below can be easily altered to test for other values of a^2. We suspect that this result is true in general but, as we said above, we do not have a proof.

In the paper "Differential Galois Theory of Difference Equations", we showed that for a difference equation $\sigma(Y) = AY$ over $C(x)$ with difference Galois group $SL_n(C)$, the differential transcendence degree is less than n^2-1 if and only if there exists a matrix B in $gl_n(C(x))$ such that $\sigma(B) = ABA^{-1} + \delta A\, A^{-1}$, where $\delta=xd/dx$. This latter equation is an $n^2 \times n^2$ system in the entries of B. In section 1, we shall derive this system for the above family of q-hypergeometric equations. In section 2, we shall derive a scalar equation for one of the entries of B.

1. The equation for B.

The matrix equation associated with our scalar equation above is $\sigma(Y) = AY$. We calculate A below

```
> with(LinearAlgebra):
```

```
> MM:=simplify(Matrix([[0,1],[ (1-x)/(a*b*x - c/q),((a+b)*x -(1+
  c/q))/(a*b*x-c/q)]]));
```

$$MM := \begin{bmatrix} 0 & 1 \\ -\frac{(-1+x)\,q}{a\,b\,x\,q-c} & \frac{x\,q\,a+x\,q\,b-q-c}{a\,b\,x\,q-c} \end{bmatrix} \tag{1}$$

```
> M1:=subs(c=q,MM);
```

$$M1 := \begin{bmatrix} 0 & 1 \\ -\frac{(-1+x)\,q}{a\,b\,x\,q-q} & \frac{x\,q\,a+x\,q\,b-2\,q}{a\,b\,x\,q-q} \end{bmatrix} \tag{2}$$

```
> A:=simplify(subs(b=a,M1));
```

$$A := \begin{bmatrix} 0 & 1 \\ -\frac{-1+x}{a^2\,x-1} & \frac{2\,(x\,a-1)}{a^2\,x-1} \end{bmatrix} \tag{3}$$

We now calculate the system associated to $\sigma(B) = ABA^{-1} + \delta A\ A^{-1}$

```
> Bt:=Matrix([[u,v],[w,z]]);
```

$$Bt := \begin{bmatrix} u & v \\ w & z \end{bmatrix} \tag{4}$$

```
> B:=Matrix([[Bt[1,1]],[Bt[1,2]],[Bt[2,1]],[Bt[2,2]]]);
```

$$B := \begin{bmatrix} u \\ v \\ w \\ z \end{bmatrix} \tag{5}$$

```
> Ct:= A.Bt.A^(-1);
```

$$Ct := \left[\left[\frac{2\,w\,(x\,a-1)}{-1+x} + z, -\frac{w\left(a^2\,x-1\right)}{-1+x}\right], \left[\frac{2\left(-\frac{(-1+x)\,u}{a^2\,x-1} + \frac{2\,(x\,a-1)\,w}{a^2\,x-1}\right)(x\,a-1)}{-1+x} - \frac{(-1+x)\,v}{a^2\,x-1} + \frac{2\,(x\,a-1)\,z}{a^2\,x-1}, -\frac{\left(-\frac{(-1+x)\,u}{a^2\,x-1} + \frac{2\,(x\,a-1)\,w}{a^2\,x-1}\right)\left(a^2\,x-1\right)}{-1+x}\right]\right] \tag{6}$$

```
> C:=Matrix([[coeff(Ct[1,1],u),coeff(Ct[1,1],v),coeff(Ct[1,1],w),
  coeff(Ct[1,1],z)],
  [coeff(Ct[1,2],u),coeff(Ct[1,2],v),coeff(Ct[1,2],w),coeff(Ct[1,
  2],z)],
  [coeff(Ct[2,1],u),coeff(Ct[2,1],v),coeff(Ct[2,1],w),coeff(Ct[2,
  1],z)],
  [coeff(Ct[2,2],u),coeff(Ct[2,2],v),coeff(Ct[2,2],w),coeff(Ct[2,
  2],z)]]);
```

$$C := \begin{bmatrix} 0 & 0 & \frac{2\,(x\,a-1)}{-1+x} & 1 \\ 0 & 0 & -\frac{a^2\,x-1}{-1+x} & 0 \\ -\frac{2\,(x\,a-1)}{a^2\,x-1} & -\frac{-1+x}{a^2\,x-1} & \frac{4\,(x\,a-1)^2}{(a^2\,x-1)\,(-1+x)} & \frac{2\,(x\,a-1)}{a^2\,x-1} \\ 1 & 0 & -\frac{2\,(x\,a-1)}{-1+x} & 0 \end{bmatrix} \qquad (7)$$

```
> ddA:= map(diff,A,x);
```

$$ddA := \begin{bmatrix} 0 & 0 \\ -\frac{1}{a^2\,x-1}+\frac{(-1+x)\,a^2}{(a^2\,x-1)^2} & \frac{2\,a}{a^2\,x-1}-\frac{2\,(x\,a-1)\,a^2}{(a^2\,x-1)^2} \end{bmatrix} \qquad (8)$$

```
> XX:=Matrix([[x,0],[0,x]]);
```

$$XX := \begin{bmatrix} x & 0 \\ 0 & x \end{bmatrix} \qquad (9)$$

```
> dA:= simplify(Multiply(XX,ddA));
```

$$dA := \begin{bmatrix} 0 & 0 \\ -\frac{x\,(-1+a^2)}{(a^2\,x-1)^2} & \frac{2\,x\,a\,(-1+a)}{(a^2\,x-1)^2} \end{bmatrix} \qquad (10)$$

```
> Et:= dA.A^(-1);
```

$$Et := \begin{bmatrix} 0 & 0 \\ -\frac{2\,x\,(-1+a^2)\,(x\,a-1)}{(a^2\,x-1)^2\,(-1+x)}+\frac{2\,x\,a\,(-1+a)}{(a^2\,x-1)^2} & \frac{x\,(-1+a^2)}{(a^2\,x-1)\,(-1+x)} \end{bmatrix} \qquad (11)$$

```
> E:= simplify(Matrix([[Et[1,1]],[Et[1,2]],[Et[2,1]],[Et[2,2]]]));
```

$$E := \begin{bmatrix} 0 \\ 0 \\ -\frac{2\,(-1+a)\,x}{(a^2\,x-1)\,(-1+x)} \\ \frac{x\,(-1+a^2)}{(a^2\,x-1)\,(-1+x)} \end{bmatrix} \qquad (12)$$

The associated system for B (now written as a colum vector as above) is $\sigma(B) = CB+E$, where C and E are as above.

2. The scalar equation for v.

To compute a scalar equation for the component v of the vector B, we proceed as follows. We will compute matrices C_i and E_i such that $\sigma^i(B) = C_i B + E_i$. Let c_i denote the entry in second row of C_i and e_i denote the entry of the second row of E_i. We will find elements y_i such that $\Sigma\ y_i\, c_i = 0$. We then will have the scalar equation $\Sigma\ y_i\, \sigma^i(v) = \Sigma\ y_i\, e_i$. Note that if e = (0,1,0,0), then for any matrix M, eM is the second row of M.

```
> e:=Matrix([[0,1,0,0]]);
```

$$e := \begin{bmatrix} 0 & 1 & 0 & 0 \end{bmatrix} \qquad \textbf{(13)}$$

```
> s1C:=subs(x=q*x,C);
```

$$s1C := \begin{bmatrix} 0 & 0 & \frac{2\,(x\,q\,a-1)}{-1+x\,q} & 1 \\ 0 & 0 & -\frac{a^2\,x\,q-1}{-1+x\,q} & 0 \\ -\frac{2\,(x\,q\,a-1)}{a^2\,x\,q-1} & -\frac{-1+x\,q}{a^2\,x\,q-1} & \frac{4\,(x\,q\,a-1)^2}{(a^2\,x\,q-1)\,(-1+x\,q)} & \frac{2\,(x\,q\,a-1)}{a^2\,x\,q-1} \\ 1 & 0 & -\frac{2\,(x\,q\,a-1)}{-1+x\,q} & 0 \end{bmatrix} \qquad \textbf{(14)}$$

```
> s2C:=subs(x=q*x,s1C);
```

$$s2C := \qquad \textbf{(15)}$$

$$\begin{bmatrix} 0 & 0 & \frac{2\,(x\,q^2\,a-1)}{-1+x\,q^2} & 1 \\ 0 & 0 & -\frac{a^2\,x\,q^2-1}{-1+x\,q^2} & 0 \\ -\frac{2\,(x\,q^2\,a-1)}{a^2\,x\,q^2-1} & -\frac{-1+x\,q^2}{a^2\,x\,q^2-1} & \frac{4\,(x\,q^2\,a-1)^2}{(a^2\,x\,q^2-1)\,(-1+x\,q^2)} & \frac{2\,(x\,q^2\,a-1)}{a^2\,x\,q^2-1} \\ 1 & 0 & -\frac{2\,(x\,q^2\,a-1)}{-1+x\,q^2} & 0 \end{bmatrix}$$

```
> s3C:=subs(x=q*x,s2C);
```

$$s3C := \qquad \textbf{(16)}$$

$$\begin{bmatrix} 0 & 0 & \frac{2\left(x q^3 a-1\right)}{-1+x q^3} & 1 \\ 0 & 0 & -\frac{a^2 x q^3-1}{-1+x q^3} & 0 \\ -\frac{2\left(x q^3 a-1\right)}{a^2 x q^3-1} & -\frac{-1+x q^3}{a^2 x q^3-1} & \frac{4\left(x q^3 a-1\right)^2}{\left(a^2 x q^3-1\right)\left(-1+x q^3\right)} & \frac{2\left(x q^3 a-1\right)}{a^2 x q^3-1} \\ 1 & 0 & -\frac{2\left(x q^3 a-1\right)}{-1+x q^3} & 0 \end{bmatrix}$$

```
> s1E:=subs(x=q*x,E);
```

$$s1E := \begin{bmatrix} 0 \\ 0 \\ -\frac{2(-1+a)xq}{\left(a^2 x q-1\right)(-1+xq)} \\ \frac{xq\left(-1+a^2\right)}{\left(a^2 x q-1\right)(-1+xq)} \end{bmatrix} \tag{17}$$

```
> s2E:=subs(x=q*x,s1E);
```

$$s2E := \begin{bmatrix} 0 \\ 0 \\ -\frac{2(-1+a)xq^2}{\left(a^2 x q^2-1\right)\left(-1+xq^2\right)} \\ \frac{xq^2\left(-1+a^2\right)}{\left(a^2 x q^2-1\right)\left(-1+xq^2\right)} \end{bmatrix} \tag{18}$$

Since $C_0 = I$, we have

```
> c0:=e:
```

Since $C_1 = C$, we have

```
> c1:= Multiply(e,C):
```

Since $C_2 = \sigma(C)C$, we have

```
> c2:=simplify(Multiply(e,Multiply(s1C,C))):
```

Since $C_2 = \sigma^2(C)\sigma(C)C$, we have

```
> c3:= simplify(Multiply(e,Multiply(s2C,Multiply(s1C,C)))):
```

A priori, there will be a dependence among e0,e1,e2,e3,e4 but in this case there is actually a dependence among e0,e1,e2,e3. To find such a dependence let

```
> MM:=Transpose(Matrix([[c0],[c1],[c2],[c3]])):
```

The dependence will be given by the following element of the nullspace of this matrix.

```
> VV:=(Transpose(op(1,NullSpace(MM))));
```

$$VV := \left[-\frac{(-1+x)(a^2 x q^2-1)(x q a-1)}{(x a-1)(a^2 x q-1)(-1+x q^2)}, \frac{(a^2 x+3+3 x^2 q a^2+x q-4 x q a-4 x a)(a^2 x q^2-1)}{(a^2 x-1)(a^2 x q-1)(-1+x q^2)}, -\frac{(a^2 x q^2-1)(x q a-1)(a^2 x+3+3 x^2 q a^2+x q-4 x q a-4 x a)}{(-1+x q^2)(a^2 x q-1)^2 (x a-1)}, 1 \right] \tag{19}$$

Letting y_{i-1} be the ith entry of this vector, we have that $\sigma^3(v)+y_2\sigma^2(v)+y_1\sigma(v)+y_0 v = e_3+y_2 e_2+y_1 e_1+y_0 e_0$. Clearly $e_0 = e_1 = 0$. A calculation shows that $E_2 = \sigma(C)E+\sigma(E)$ and $E_2 = \sigma^2(C)\sigma(C)E+\sigma^2(C)\sigma(E)+\sigma^2(E)$ so

```
> e0:=0;
```

$$e0 := 0 \tag{20}$$

```
> e1:=(e.E)[1,1];
```

$$e1 := 0 \tag{21}$$

```
> e2:=(Multiply(e,Multiply(s1C,E)+s1E))[1,1];
```

$$e2 := \frac{2(a^2 x q-1)(-1+a)x}{(-1+x q)(a^2 x-1)(-1+x)} \tag{22}$$

```
> e3:=Multiply(e,(Multiply(s2C,Multiply(s1C,E))+Multiply(s2C,s1E)+
  s2E))[1,1];
```

$$e3 := -\frac{1}{-1+x q^2}\left((a^2 x q^2-1)\left(-\frac{8(x q a-1)^2(-1+a)x}{(a^2 x q-1)(-1+x q)(a^2 x-1)(-1+x)} + \frac{2(x q a-1)x(-1+a^2)}{(a^2 x q-1)(a^2 x-1)(-1+x)}\right)\right) + \frac{2(a^2 x q^2-1)(-1+a)x q}{(-1+x q^2)(a^2 x q-1)(-1+x q)} \tag{23}$$

```
> EE:=Matrix([[e0],[e1],[e2],[e3]]);
```

$$EE := \left[\left[\begin{array}{c} 0 \end{array} \right], \left[\begin{array}{c} 0 \end{array} \right], \left[\begin{array}{c} \frac{2(a^2 x q-1)(-1+a)x}{(-1+x q)(a^2 x-1)(-1+x)} \end{array} \right], \left[\begin{array}{c} -\frac{1}{-1+x q^2}\left((a^2 x q^2-1)\left(-\frac{8(x q a-1)^2(-1+a)x}{(a^2 x q-1)(-1+x q)(a^2 x-1)(-1+x)} + \frac{2(x q a-1)x(-1+a^2)}{(a^2 x q-1)(a^2 x-1)(-1+x)}\right)\right) + \frac{2(a^2 x q^2-1)(-1+a)x q}{(-1+x q^2)(a^2 x q-1)(-1+x q)} \end{array} \right] \right] \tag{24}$$

The following WWW will be the left hand side of $\sigma^3(v) + y_2\sigma^2(v) + y_1\sigma(v) + y_0 v = e_3 + y_2 e_2 + y_1 e_1 + y_0 e_0$ and the entries of VV will be the coefficients of the the right hand side.

```
> WWW:=simplify(VV.EE);
```

$$WWW := \left[\frac{2\left(x^2 q^2 a^2 + x^2 q a^4 - x^2 q^2 a^3 - x^2 q a^3 + a - a^2 - q + q a\right)\left(a^2 x q^2 - 1\right) x}{(-1 + x q)\left(a^2 x q - 1\right)\left(a^2 x - 1\right)\left(-1 + x q^2\right)(x a - 1)} \right] \tag{25}$$

```
> VV;
```

$$\left[-\frac{(-1+x)\left(a^2 x q^2 - 1\right)(x q a - 1)}{(x a - 1)\left(a^2 x q - 1\right)\left(-1 + x q^2\right)}, \frac{\left(a^2 x + 3 + 3 x^2 q a^2 + x q - 4 x q a - 4 x a\right)\left(a^2 x q^2 - 1\right)}{\left(a^2 x - 1\right)\left(a^2 x q - 1\right)\left(-1 + x q^2\right)}, -\frac{\left(a^2 x q^2 - 1\right)(x q a - 1)\left(a^2 x + 3 + 3 x^2 q a^2 + x q - 4 x q a - 4 x a\right)}{\left(-1 + x q^2\right)\left(a^2 x q - 1\right)^2 (x a - 1)}, 1 \right] \tag{26}$$

We now lcear the denominators and produce the third order equation EQN for v(x).

```
> W:=denom(WWW[1])*(a^2*x*q-1)*WWW;
```

$$W := \left[2\left(a^2 x q - 1\right)\left(x^2 q^2 a^2 + x^2 q a^4 - x^2 q^2 a^3 - x^2 q a^3 + a - a^2 - q + q a\right)\left(a^2 x q^2 - 1\right) x\right] \tag{27}$$

```
> V:= denom(WWW[1])*(a^2*x*q-1)*VV;
```

$$V := \left[-(-1 + x q)\left(a^2 x q - 1\right)\left(a^2 x - 1\right)(-1 + x)\left(a^2 x q^2 - 1\right)(x q a - 1), (-1 + x q)\left(a^2 x q - 1\right)(x a - 1)\left(a^2 x + 3 + 3 x^2 q a^2 + x q - 4 x q a - 4 x a\right)\left(a^2 x q^2 - 1\right), -(-1 + x q)\left(a^2 x - 1\right)\left(a^2 x q^2 - 1\right)(x q a - 1)\left(a^2 x + 3 + 3 x^2 q a^2 + x q - 4 x q a - 4 x a\right), (-1 + x q)\left(a^2 x q - 1\right)^2\left(a^2 x - 1\right)\left(-1 + x q^2\right)(x a - 1) \right] \tag{28}$$

```
> y:= Matrix([[v(x),v(q*x),v(q^2*x),v(q^3*x)]]);
```

$$y := \left[\begin{array}{cccc} v(x) & v(x q) & v(x q^2) & v(x q^3) \end{array} \right] \tag{29}$$

```
> EQN:= (V.Transpose(y))[1] = W[1];
```

$$EQN := -(-1 + x q)\left(a^2 x q - 1\right)\left(a^2 x - 1\right)(-1 + x)\left(a^2 x q^2 - 1\right)(x q a - 1) v(x) + (-1 + x q)\left(a^2 x q - 1\right)(x a - 1)\left(a^2 x + 3 + 3 x^2 q a^2 + x q - 4 x q a - 4 x a\right)\left(a^2 x q^2 - 1\right) v(x q) - (-1 + x q)\left(a^2 x - 1\right)\left(a^2 x q^2 - 1\right)(x q a - 1)\left(a^2 x + 3 + 3 x^2 q a^2 + x q - 4 x q a - 4 x a\right) v\left(x q^2\right) + (-1 + x q)\left(a^2 x q - 1\right)^2\left(a^2 x - 1\right)\left(-1 + x q^2\right)(x a - 1) v\left(x q^3\right) = 2\left(a^2 x q - 1\right)\left(x^2 q^2 a^2 + x^2 q a^4 - x^2 q^2 a^3 - x^2 q a^3 + a - a^2 - q + q a\right)\left(a^2 x q^2 - 1\right) x \tag{30}$$

We wish to show that this equation has no nonzero rational solutions. In order to use MAPLE taking into account that a^2 is a power of q. To do this we will replace q by q^2 and consider a sixth order equation.

This avoids haing to deal with the square root of q when a^2 is an odd power of q. We then apply the MAPLE command RationalSolution. This command will return an empty set when there are no rational solutions.

```
> EQN0:=subs(q=t,EQN);
```

$$EQN0 := -(-1+xt)\left(a^2xt-1\right)\left(a^2x-1\right)(-1+x)\left(a^2xt^2-1\right)(xta-1)v(x) + (-1+xt)\left(a^2xt-1\right)(xa-1)\left(a^2x+3+3x^2ta^2+xt-4xta-4xa\right)\left(a^2xt^2-1\right)v(xt) - (-1+xt)\left(a^2x-1\right)\left(a^2xt^2-1\right)(xta-1)\left(a^2x+3+3x^2ta^2+xt-4xta-4xa\right)v\left(xt^2\right) + (-1+xt)\left(a^2xt-1\right)^2\left(a^2x-1\right)\left(-1+xt^2\right)(xa-1)v\left(xt^3\right) = 2\left(a^2xt-1\right)\left(x^2t^2a^2+x^2ta^4-x^2t^2a^3-x^2ta^3+a-a^2-t+ta\right)\left(a^2xt^2-1\right)x \quad (31)$$

```
> EQN1:=subs(a=q^1,t=q^2,EQN0);
```

$$EQN1 := -\left(-1+xq^2\right)^2\left(q^4x-1\right)(-1+x)\left(q^6x-1\right)\left(-1+xq^3\right)v(x) + \left(-1+xq^2\right)\left(q^4x-1\right)(-1+xq)\left(2xq^2+3+3x^2q^4-4xq^3-4xq\right)\left(q^6x-1\right)v\left(xq^2\right) - \left(-1+xq^2\right)^2\left(q^6x-1\right)\left(-1+xq^3\right)\left(2xq^2+3+3x^2q^4-4xq^3-4xq\right)v\left(q^4x\right) + \left(-1+xq^2\right)^2\left(q^4x-1\right)^3(-1+xq)v\left(q^6x\right) = 2\left(q^4x-1\right)\left(2x^2q^6-x^2q^7-x^2q^5+q-2q^2+q^3\right)\left(q^6x-1\right)x \quad (32)$$

```
> with(QDifferenceEquations):
```

The MAPLE command RationalSolution will yield a basis of the rational solutions of a q-difference equation. When the output is empty, we can conclude that there are no nonzero solutions.

```
> sol1 := RationalSolution(EQN1, v(x), {}, output = basis[_K]);
```

$$sol1 := \quad (33)$$

```
> EQN3:=subs(a=q^3,t=q^2,EQN0):
  with(QDifferenceEquations):
  sol1 := RationalSolution(EQN3, v(x), {}, output = basis[_K]);
```

$$sol1 := \quad (34)$$

```
> EQN5:=subs(a=q^5,t=q^2,EQN0):
  with(QDifferenceEquations):
  sol1 := RationalSolution(EQN5, v(x), {}, output = basis[_K]);
```

$$sol1 := \quad (35)$$

```
> EQNminus1:=subs(a=q^(-1),t=q^2,EQN0):
  with(QDifferenceEquations):
  sol1 := RationalSolution(EQNminus1, v(x), {}, output = basis[_K])
  ;
```

$$sol1 := \quad (36)$$

```
> EQNminus3:=subs(a=q^(-3),t=q^2,EQN0):
  with(QDifferenceEquations):
  sol1 := RationalSolution(EQNminus3, v(x), {}, output = basis[_K])
  ;
```

$$sol1 := \quad (37)$$

```
> EQNminus5:=subs(a=q^(-5),t=q^2,EQN0):
  with(QDifferenceEquations):
  sol1 := RationalSolution(EQNminus5, v(x), {}, output = basis[_K])
  ;
```

$$sol1 := \tag{38}$$

```
> EQN2:=subs(a=-q,t=q,EQN0):
  with(QDifferenceEquations):
  sol1 := RationalSolution(EQN2, v(x), {}, output = basis[_K]);
```

$$sol1 := \tag{39}$$